
eAppendix 
 

/*Generate data*/ 

DATA a; 

 CALL STREAMINIT(123); 

 DO id=1 to 1000000; 

 

  PA=0.5; 

  A=rand("Bern",PA); 

   

  PU = 0.3; 

  U=rand("Bern",PU); 

 

  PW = 0.2+0.4*A+0.3*U; 

  W = rand("Bern", PW); 

 

  PS = 0.1+0.8*W; 

  S = rand("Bern", PS); 

 

  PY=0.3+0.5*U-0.2*A; 

  Y=rand("Bern",PY); 

 

  OUTPUT; 

 END; 

RUN; 

 

/*Truth*/ 

PROC GENMOD DATA=a DESC; 

 MODEL Y=A / DIST=bin LINK=identity; 

 ODS SELECT modelinfo parameterestimates; 

 TITLE "truth"; 

RUN; 

 

/*Crude*/ 

PROC GENMOD DATA=a DESC; 

 MODEL Y=A / DIST=bin LINK=identity; 

 ODS SELECT modelinfo parameterestimates; 

 TITLE "crude"; 

 WHERE S=0; 

RUN; 

 

/*Stratified on W*/ 

PROC GENMOD DATA=a DESC; 

 MODEL Y=A W/ DIST=bin LINK=identity; 

 ODS SELECT modelinfo parameterestimates; 

 TITLE "stratified"; 

 WHERE S=0; 

RUN; 

 

/*Gformula*/ 

TITLE "G-comp Models"; 

ODS SELECT NONE; 

PROC GENMOD DATA=a DESC; 

 WHERE S=0; 

 MODEL y=a w / DIST=bin LINK=identity; 

 ODS OUTPUT parameterestimates=yest(keep=parameter estimate stderr); 



RUN; 

 

PROC GENMOD DATA=a DESC; 

 MODEL W=A / DIST=bin LINK=identity; 

 ODS OUTPUT parameterestimates=west(keep=parameter estimate stderr); 

RUN; 

ODS SELECT ALL; 

 

DATA west;  

 SET west;  

 RETAIN z 1 intw intwse; 

 IF parameter="Intercept" THEN DO; intw=estimate; intwse=stderr; END; 

 IF parameter="A" THEN DO; wona=estimate; wonase=stderr; OUTPUT; END; 

 KEEP z intw intwse wona wonase; 

RUN; 

 

DATA yest;  

 SET yest;  

 RETAIN z 1 inty intyse yona yonase; 

 IF parameter="Intercept" THEN DO; inty=estimate; intyse=stderr; END; 

 IF parameter="A" THEN DO; yona=estimate; yonase=stderr; END; 

 IF parameter="W" THEN DO; yonw=estimate; yonwse=stderr; OUTPUT; END; 

 KEEP z inty intyse yona yonase yonw yonwse; 

RUN; 

 

PROC PRINT DATA=west; TITLE "G-comp model P(W|A)"; 

PROC PRINT DATA=yest; TITLE "G-comp model P(Y|A,W,S=0)"; RUN; 

 

DATA b; 

 MERGE west yest; 

 BY z; 

 gcomp1=0; gcomp0=0; 

 DO A=0 TO 1; 

  DO W=0 TO 1; 

   pw = (intw + wona*A); 

   IF W=0 THEN pw=1-pw; 

   IF A=0 THEN gcomp0=gcomp0 + pw*(inty + yona*A + yonw*W); 

   ELSE gcomp1=gcomp1 + pw*(inty + yona*A + yonw*W); 

  END; 

 END; 

 gcomp=gcomp1-gcomp0; 

 DROP A W pw; 

RUN; 

 

PROC PRINT DATA = b; VAR gcomp; TITLE "gcomp"; RUN; 

 


